C02AHF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

C02AHF determines the roots of a quadratic equation with complex coefficients.

2 Specification

```
SUBROUTINE C02AHF(AR, AI, BR, BI, CR, CI, ZSM, ZLG, IFAIL)
INTEGER           IFAIL
real              AR, AI, BR, BI, CR, CI, ZSM(2), ZLG(2)
```

3 Description

The routine attempts to find the roots of the quadratic equation $az^2 + bz + c = 0$ (where a, b and c are complex coefficients), by carefully evaluating the 'standard' closed formula

$$z = \frac{-b \pm \sqrt{b^2 - 4ac}}{2a}.$$

It is based on the routine CQDRTC from Smith [1].

Note. It is not necessary to scale the coefficients prior to calling the routine.

4 References

[1] Smith B T (1967) ZERPOL: A zero finding algorithm for polynomials using Laguerre's method *Technical Report* Department of Computer Science, University of Toronto, Canada

5 Parameters

1: AR — ***real*** *Input*

2: AI — ***real*** *Input*

On entry: AR and AI must contain the real and imaginary parts respectively of a, the coefficient of z^2.

3: BR — ***real*** *Input*

4: BI — ***real*** *Input*

On entry: BR and BI must contain the real and imaginary parts respectively of b, the coefficient of z.

5: CR — ***real*** *Input*

6: CI — ***real*** *Input*

On entry: CR and CI must contain the real and imaginary parts respectively of c, the constant coefficient.

7: ZSM(2) — ***real*** array *Output*

On exit: the real and imaginary parts of the smallest root in magnitude are stored in ZSM(1) and ZSM(2) respectively.

8: ZLG(2) — ***real*** array *Output*

On exit: the real and imaginary parts of the largest root in magnitude are stored in ZLG(1) and ZLG(2) respectively.

9: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

On entry, (AR,AI) = (0, 0). In this case, ZSM(1) and ZSM(2) contain the real and imaginary parts respectively of the root $-c/b$.

IFAIL = 2

On entry, (AR,AI) = (0, 0) and (BR,BI) = (0, 0). In this case, ZSM(1) contains the largest machine representable number (see X02ALF) and ZSM(2) contains zero.

IFAIL = 3

On entry, (AR,AI) = (0, 0) and the root $-c/b$ overflows. In this case, ZSM(1) contains the largest machine representable number (see X02ALF) and ZSM(2) contains zero.

IFAIL = 4

On entry, (CR,CI) = (0, 0) and the root $-b/a$ overflows. In this case, both ZSM(1) and ZSM(2) contain zero.

IFAIL = 5

On entry, $\tilde{b}$ is so large that $\tilde{b}^2$ is indistinguishable from $\tilde{b}^2 - 4\tilde{a}\tilde{c}$ and the root $-b/a$ overflows, where $\tilde{b} = \max(|\text{BR}|, |\text{BI}|)$, $\tilde{a} = \max(|\text{AR}|, |\text{AI}|)$ and $\tilde{c} = \max(|\text{CR}|, |\text{CI}|)$. In this case, ZSM(1) and ZSM(2) contain the real and imaginary parts respectively of the root $-c/b$.

If IFAIL > 0 on exit, then ZLG(1) contains the largest machine representable number (see X02ALF) and ZLG(2) contains zero.

7 Accuracy

If IFAIL = 0 on exit, then the computed roots should be accurate to within a small multiple of the ***machine precision*** except when underflow (or overflow) occurs, in which case the true roots are within a small multiple of the underflow (or overflow) threshold of the machine.

8 Further Comments

None.

9 Example

To find the roots of the quadratic equation $z^2 - (3.0 - 1.0i)z + (8.0 + 1.0i) = 0$.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     C02AHF Example Program Text
*     Mark 14 Release.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             AI, AR, BI, BR, CI, CR
      INTEGER          IFAIL
*     .. Local Arrays ..
      real             ZLG(2), ZSM(2)
*     .. External Subroutines ..
      EXTERNAL         C02AHF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'C02AHF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) AR, AI, BR, BI, CR, CI
      IFAIL = 0
*
      CALL C02AHF(AR,AI,BR,BI,CR,CI,ZSM,ZLG,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Roots of quadratic equation'
      WRITE (NOUT,*)
      WRITE (NOUT,99999) 'z = ', ZSM(1), ZSM(2), '*i'
      WRITE (NOUT,99999) 'z = ', ZLG(1), ZLG(2), '*i'
      STOP
*
99999 FORMAT (1X,A,1P,e12.4,SP,e14.4,A)
      END
```

9.2 Program Data

```
C02AHF Example Program Data
 1.0    0.0   -3.0    1.0    8.0    1.0       :AR  AI  BR  BI  CR  CI
```

9.3 Program Results

```
 C02AHF Example Program Results

 Roots of quadratic equation

 z =   1.0000E+00   +2.0000E+00*i
 z =   2.0000E+00   -3.0000E+00*i
```